

Need a Scientific Journal Ready Graphic? No Problem!

Charlotte Baker, Florida Agricultural and Mechanical University

ABSTRACT

Graphics are an excellent way to display results from multiple statistical analyses and get a visual message across to the correct audience. Scientific journals often have very precise requirements for graphs that are submitted with manuscripts. While authors often find themselves using tools other than SAS® to create these graphs, the combination of the SGPLOT procedure and the Output Delivery System enables authors to create what they need in the same place as they conducted their analysis. This presentation focuses on two methods for creating a publication quality graphic in SAS® 9.4 and provides solutions for some issues encountered when doing so.

INTRODUCTION

The graphical presentation of data is both an art and a science. Often, researchers are required to present results of studies in a manner that is clear not only to other researchers but also to the general public. Graphics such as line graphs and bar charts can be very useful as tools to accomplish this. One caveat of graphical display is that the graph or chart needs to be readable if expanded for display in a presentation or minimized if used in a print journal. It also needs to look good on websites and in PDF documents. While there are many programs that can be used to create graphics, SAS contains tools that can be combined to do a more than sufficient job, namely the statistical graphics (SG) procedures and the Output Delivery System (ODS).

There are many reasons why one would choose to use a single program to complete both statistical analysis and the graphing of that data. However, it is arguably more important that SAS's SG procedures can create many different types of graphs based on many different statistics and allows the complete customization of the output. This means that no matter where the graphic needs to be used, it will reflect exactly what the researcher wanted someone to see. What program is used to make the graphic matters quite a bit when journals typically provide resolution, color, and/or size criteria for graphics submitted along with manuscripts. The evolution of the SG procedures along with ODS means that SAS can create graphics that can be used for such purposes. This paper will demonstrate two methods of creating a publication quality graphic using PROC SGPLOT.

SCENARIO

You are a researcher in a team that has completed a study of stroke hospitalizations due to sickle cell disease over a period of six data collection periods. We are interested in graphing information to show whether the proportion of stroke hospitalizations for children with sickle cell disease has changed over the data collection periods for four age groups. Our four age groups are 1-4 years, 5-9 years, 10-14 years, and 15-18 years. To find out whether there was a trend in the proportion for each age group, we used a linear regression. We want to create a single line graph (see Figure 1) with a regression trend line for each age group and a marker for the proportion of hospitalizations for each age group by year.

METHOD 1 - CODING

CREATING THE IMAGE

You need a graphic that can be used quickly for a slide presentation. The screen for the presentation is large but the factor of most concern is the file size so that the presentation loads quickly. You choose to create a TIFF image.




Figure 1. Presentation Quality Graphic – TIFF at 1200 DPI (Baker et al)

The code to create a basic graph for this scenario requires information for the x-axis and the y-axis. Because we want to have a linear regression line to show the trend, we need to use a REG statement. The REG statement fits a regression line to your data.

```
ODS GRAPHICS ON;
PROC SGPLOT DATA = trenddata;
REG X=year Y=ageper1;
REG X=year Y=ageper2;
REG X=year Y=ageper3;
REG X=year Y=ageper4;
XAXIS DISCRETEORDER=DATA VALUES=(1997 2000 2003 2006 2009 2012);
YAXIS VALUES =(00 05 10 15 20 25 30 35 40 45 50 55 60);
RUN;
```

The SG procedures use ODS GRAPHICS to provide the best image possible so this feature needs to be turned on. Depending on your SAS installation, ODS GRAPHICS may be on by default. Because we are trying to plot four separate lines (one for each group), this code shows four separate REG statements. In each REG statement, “year” is the name of the variable that contains the information that should be on the x-axis. In this case, we are plotting the information on the y-axis by year and the variable for year is “year”. The first line we want to plot is for the first age group (1-4 years of age). We are plotting the proportion of stroke hospitalizations due to sickle cell disease for that age group on the y-axis and the variable that represents this proportion is “ageper1”. The XAXIS statement allows us to specify the options for what the x-axis looks like. The DISCRETEORDER option specifies the order of the tick marks on the x-axis. In this case we want them in the order of the data. The values we want to appear (instead

of the default values) are the actual years of data collection that we studied – 1997, 2000, 2003, 2006, 2009, and 2012. The YAXIS statement allows us to specify the options for how the y-axis looks. Because we wanted to start our axis at 0 and because we were familiar with the age proportions that we were plotting on the y-axis, we made the values for the y-axis range from 0 to 60 in increments of 5. Each of these values will have a tick-mark on the y-axis.

While the code above will provide a graph that can certainly be used for presentation, it is often a goal to be sure that whoever is viewing the graph can decipher the different items being plotted. Because of this, we want to be sure to be more specific about how each of the linear regression lines look and how the markers for each proportion for each age look. We do this with additional options added to the REG statements.

```
PROC SGPLOT DATA = trenddata;
  REG X=year Y=ageper1 / LINEATTRS = (COLOR = BLACK PATTERN = DASHDOTDOT)
  MARKERATTRS = (SYMBOL = TRIANGLEFILLED COLOR = BLACK) LEGENDLABEL = "1-4
  Years of Age";
  REG X=year Y=ageper2 / LINEATTRS = (COLOR = BLACK PATTERN = DASH)
  MARKERATTRS = (SYMBOL = CIRCLEFILLED COLOR = BLACK) LEGENDLABEL = "5-9
  Years of Age";
  REG X=year Y=ageper3 / LINEATTRS = (COLOR = BLACK PATTERN = DOT)
  MARKERATTRS = (SYMBOL = SQUAREFILLED COLOR = BLACK) LEGENDLABEL = "10-14
  Years of Age";
  REG X=year Y=ageper4 / LINEATTRS = (COLOR = BLACK PATTERN = LONGDASH)
  MARKERATTRS = (SYMBOL = STARFILLED COLOR = BLACK) LEGENDLABEL = "15-18
  Years of Age";
  XAXIS DISCRETEORDER=DATA VALUES=(1997 2000 2003 2006 2009 2012);
  YAXIS VALUES =(00 05 10 15 20 25 30 35 40 45 50 55 60);
RUN;
```

To add options to the REG statement (or any specific graph statement in PROC SGPLOT), insert a forward slash after identifying the X and Y variables and follow it with the options you want. In this study, we wanted to change the color and pattern of each line, the shape and color of each marker that identified the proportions, and add a specific label for the graph's legend. LINEATTRS is the option that changes the attributes of the line you see in the graph. Because we are changing more than one attribute (color and pattern), we need to put the attributes we are changing inside parentheses. To specify colors in SAS/GRAPH®, one has several options including RGB, CMYK, and SAS Registry Color options. In this example, we used a recognized SAS Registry Color – black. Just as there are many options to change the color of a line, there are many options to change the pattern of the lines that appear for our REG line. We selected to use a LONGDASH pattern for the fourth regression line.

The markers in this example are the items that appear on the graph to indicate where on the graph the proportion for a specific age group in a specific year. We change what the markers look like by using the MARKERATTRS option. In this example, we wanted to change the shape of the marker (done with the SYMBOL option) and the color of that marker. Just like the color for the LINEATTRS option, we change the color for the MARKERATTRS option using a SAS registry color. Because the fourth REG statement is for the 15-18 year old age group, each marker on the graph for that group will be a filled in star shape (STARFILLED). There are approximately two dozen marker symbols to choose from. See the reference section for links to further examples and details for specifying colors, patterns, and markers.

If you were to print this graph using the code above, you would have a graph that has everything you want in the graph but it is missing labels for the axes and the fonts for the labels and axes will be smaller than is typically warranted when making graphics for professional purposes. To correct this, we add more options to the XAXIS and YAXIS statements. The graph also does not have a legend to explain to us what the lines and markers mean. To add one, we add the KEYLEGEND statement.

```
PROC SGPLOT DATA = trenddata;
  REG X=year Y=ageper1 / LINEATTRS = (COLOR = BLACK PATTERN = DASHDOTDOT)
  MARKERATTRS = (SYMBOL = TRIANGLEFILLED COLOR = BLACK) LEGENDLABEL = "1-4
  Years of Age";
```

```

REG X=year Y=ageper2 / LINEATTRS = (COLOR = BLACK PATTERN = DASH)
MARKERATTRS = (SYMBOL = CIRCLEFILLED COLOR = BLACK) LEGENDLABEL = "5-9
Years of Age";
REG X=year Y=ageper3 / LINEATTRS = (COLOR = BLACK PATTERN = DOT)
MARKERATTRS = (SYMBOL = SQUAREFILLED COLOR = BLACK) LEGENDLABEL = "10-14
Years of Age";
REG X=year Y=ageper4 / LINEATTRS = (COLOR = BLACK PATTERN = LONGDASH)
MARKERATTRS = (SYMBOL = STARFILLED COLOR = BLACK) LEGENDLABEL = "15-18
Years of Age";
XAXIS VALUEATTRS = (FAMILY = ARIAL SIZE = 18pt) DISCRETEORDER=DATA
VALUES=(1997 2000 2003 2006 2009 2012) LABEL = "Year" LABELATTRS = (SIZE
=18pt);
YAXIS VALUEATTRS = (FAMILY = ARIAL SIZE = 18pt) LABEL = "Percent (%) of
Stroke Discharges by Age" LABELATTRS = (SIZE =18pt) VALUES =(00 05 10 15 20
25 30 35 40 45 50 55 60);
KEYLEGEND / NOBORDER LOCATION = OUTSIDE POSITION = BOTTOMLEFT VALUEATTRS =
(FAMILY = arial SIZE = 18pt);
RUN;

```

The **VALUEATTRS** option allows us to change the attributes of the tick mark value labels for each axis. For the **Y**AXIS, we specified that any text for the y-axis should be in 18-point Arial font. In this example, we chose 18-point font because that is the smallest recommended font size for anything presented in a slideshow. SAS accepts font sizes to be specified in point size, pixels, inches, centimeters, millimeters, or percentages. In order for the font selected in the **FAMILY** option to actually appear in the graph, it must be a TrueType font and must be present on the computer being used to make the graph. The **LABEL** option is where you can specify what text should appear on the axis. For the x-axis, we decided to simply write "Year". By using the **LABELATTRS** option, we specified that "Year" should also appear in 18-point font. **KEYLEGEND** specifies that we need a legend for our graph. In this case, we did not want a border for the legend and wanted it outside the main graph area. We wanted it to appear instead to the bottom left outside the graph. We wanted the information in the key to appear in Arial 18-point font.

We are now ready to output this graph as a TIFF image as we decided in the beginning. To do this, the HTML output needs to be turned off and the LISTING output needs to be turned on. This is very important as the default in current SAS 9.4 installations is for HTML to be turned on. In addition, one must specify where the graph should be saved. The **GPATH** option for **ODS LISTING** allows us to do this.

```

ODS HTML OFF;
ODS LISTING GPATH = "C:\tiff";

```

Without further specifications, an image will be saved but it will be in PNG format. To change the image format, we need to add the **IMAGEFMT** option to the **ODS GRAPHICS** statement.

```

ODS HTML OFF;
ODS LISTING GPATH = "C:\tiff";
ODS GRAPHICS ON / IMAGEFMT = TIFF;
PROC SGPLOT DATA = trenddata;
REG X=year Y=ageper1 / LINEATTRS = (COLOR = BLACK PATTERN = DASHDOTDOT)
MARKERATTRS = (SYMBOL = TRIANGLEFILLED COLOR = BLACK) LEGENDLABEL = "1-4
Years of Age";
REG X=year Y=ageper2 / LINEATTRS = (COLOR = BLACK PATTERN = DASH)
MARKERATTRS = (SYMBOL = CIRCLEFILLED COLOR = BLACK) LEGENDLABEL = "5-9
Years of Age";
REG X=year Y=ageper3 / LINEATTRS = (COLOR = BLACK PATTERN = DOT)
MARKERATTRS = (SYMBOL = SQUAREFILLED COLOR = BLACK) LEGENDLABEL = "10-14
Years of Age";
REG X=year Y=ageper4 / LINEATTRS = (COLOR = BLACK PATTERN = LONGDASH)
MARKERATTRS = (SYMBOL = STARFILLED COLOR = BLACK) LEGENDLABEL = "15-18
Years of Age";

```

```

XAXIS VALUEATTRS = (FAMILY = ARIAL SIZE = 18pt) DISCRETEORDER=DATA
VALUES=(1997 2000 2003 2006 2009 2012) LABEL = "Year" LABELATTRS = (SIZE
=18pt);
YAXIS VALUEATTRS = (FAMILY = ARIAL SIZE = 18pt) LABEL = "Percent (%) of
Stroke Discharges by Age" LABELATTRS = (SIZE =18pt) VALUES =(00 05 10 15 20
25 30 35 40 45 50 55 60);
KEYLEGEND / NOBORDER LOCATION = OUTSIDE POSITION = BOTTOMLEFT VALUEATTRS =
(FAMILY = arial SIZE = 18pt);
RUN;

```

We now have a graphic that we can use in our presentation!

CHANGING THE DPI OF THE IMAGE

After your presentation, you decide to submit your study results for publication in a manuscript that requires the image to be a TIFF image that is 1200DPI. Your current method of creating a TIFF image will not create an image with that type of resolution (the default is 100 DPI in HTML). The alternatives are to use the IMAGE_DPI option for ODS or the DPI option for ODS. IMAGE_DPI will work in ODS LISTING or ODS HTML to create images but neither allows for the creation of a TIFF image. Because I could use a free program outside of SAS to convert the PNG image that DPI and IMAGE_DPI could provide, I elected to create a PNG image. In addition to creating the image in HTML, I elected to create it in a PDF and an RTF file as well.

```

ODS HTML DPI = 1200;
ODS PDF DPI = 1200 FILE = "C:\tiff\graph.pdf";
ODS RTF DPI = 1200 FILE = "C:\tiff\graph.doc";
PROC SGPLOT DATA = trenddata;
REG X=year Y=ageper1 / LINEATTRS = (COLOR = BLACK PATTERN = DASHDOTDOT)
MARKERATTRS = (SYMBOL = TRIANGLEFILLED COLOR = BLACK) LEGENDLABEL = "1-4
Years of Age";
REG X=year Y=ageper2 / LINEATTRS = (COLOR = BLACK PATTERN = DASH)
MARKERATTRS = (SYMBOL = CIRCLEFILLED COLOR = BLACK) LEGENDLABEL = "5-9
Years of Age";
REG X=year Y=ageper3 / LINEATTRS = (COLOR = BLACK PATTERN = DOT)
MARKERATTRS = (SYMBOL = SQUAREFILLED COLOR = BLACK) LEGENDLABEL = "10-14
Years of Age";
REG X=year Y=ageper4 / LINEATTRS = (COLOR = BLACK PATTERN = LONGDASH)
MARKERATTRS = (SYMBOL = STARFILLED COLOR = BLACK) LEGENDLABEL = "15-18
Years of Age";
XAXIS VALUEATTRS = (FAMILY = ARIAL SIZE = 12pt) DISCRETEORDER=DATA
VALUES=(1997 2000 2003 2006 2009 2012) LABEL = "Year" LABELATTRS = (SIZE
=12pt);
YAXIS VALUEATTRS = (FAMILY = ARIAL SIZE = 12pt) LABEL = "Percent (%) of
Stroke Discharges by Age" LABELATTRS = (SIZE =12pt) VALUES =(00 05 10 15 20
25 30 35 40 45 50 55 60);
KEYLEGEND / NOBORDER LOCATION = OUTSIDE POSITION = BOTTOMLEFT VALUEATTRS =
(FAMILY = arial SIZE = 18pt);
RUN;
ODS PDF CLOSE;
ODS RTF CLOSE;

```

To change the DPI, one can use the DPI option and add a value that it should be. In this example, DPI = 1200 was the way to change the DPI of the image to 1200 DPI. Be sure to close the PDF and RTF ODS locations.

A serious problem generally occurs when trying to select a large DPI. SAS will quickly stop processing and provide the following error in the log:

```
Java virtual machine exception. java.lang.OutOfMemoryError: Java heap
space. NOTE: The SAS System stopped processing this step because of errors.
```


This message means that the java virtual machine in SAS does not have enough memory available to use to create the graphic. You can give it more memory to use by making changes in the root folder of your computer where SAS is installed. Directions to solve this problem are available from the SAS Knowledge Base (see references). By following these directions, you can give SAS access to more memory to use from your computer and it will be able to create the graphic you want to see. Any missteps in implementing the directions from that page can result in your computer or SAS becoming unusable so be sure that you're comfortable performing the procedure before attempting.

Once your SAS installation has enough Java Heap Space, try the same syntax again and enjoy the result! (Figure 1)

METHOD 2 – ODS GRAPHICS DESIGNER

The ODS Graphics Designer is a menu driven tool that can be used to create a graphic with little to no code manipulation. It is available in several versions of SAS including SAS 9.4. It will provide the code that was used to create the graphic so one can continue to alter the graphic beyond the basic capabilities of the program. This will be necessary to obtain our final 1200 DPI image. To make the graph we previously made using code in Method 1, we will need to make and overlay eight different graphs; each age group will need one regression plot and one scatter plot. All values selected in the menus will be the same selections we previously made using code.

Before using the ODS Graphics Designer, be sure to create your data set in SAS 9.4 (or whatever version you are using) or to run a LIBNAME statement to tell SAS where your data is that you want to use to make your graphic. This will allow you to select the appropriate data set from the menus. To open the ODS Graphics Designer within SAS 9.4, select TOOLS on the menu bar then select ODS Graphics Designer (Display 1).




Display 1. Opening ODS Graphics Designer


From the splash screen that opens (Display 2), we have the option to select a graph right away or we can create a blank canvas to work from. Because we know we need a scatter plot, we can double click on the SCATTER PLOT in the Graph Gallery. In the ASSIGN DATA window that opens, we need to select the location of our data and the data we would like to use. In the PLOT VARIABLES tab, we should select “year” for the X variable and “ageper1” for the Y variable. We can name our scatter plot. In Display 3, you will see that this first plot was renamed “scatter age group 1”. These names are for your use and do not appear on your graphic. They are useful when attempting to remember which plot is which and you have

several but are not required. After you are satisfied with the information you have entered, select OK. Your data is now plotted on the graph.

To set our own values for the labels for the X and Y axes and to set the values for the tick marks, we need to edit the plot properties. To do this, right click on the graph and select PLOT PROPERTIES from the drop down menu. The PLOT PROPERTIES menu allows us to change values for every graph we




Display 2. ODS Graphics Designer Initial Screen




Display 3. ODS Graphics Designer Assign Data to Scatter Plot One

create from a single place. We can set the markers for the points on our scatter plots, the colors of the lines, the size of the markers, and provide a label for each plot for the legend (Display 4). As we add more plots to this graph, we can choose which graph to edit from the PLOT dropdown menu. For this first scatter plot, select the color black, the triangle filled symbol, change to SIZE to 19, and provide the legend label "AgePer1".




Display 4. Plot Properties Change Marker Values



Display 5. Plot Properties Change the Range of X-Axis Values


If you select the AXES tab at the top of the plot properties window, you will see the options to edit the X-axis (Display 5). In the box that appears, write “Year” for the LABEL to change the label on the X-axis then select the LABEL tab under AXES (see arrow in Display 5). Change the FONT SIZE to 18. The X-axis label will now be 18-point font. Select the VALUE tab under AXES and change the FONT SIZE to 18. Any tick mark values on the X-axis will now be 18-point font. Select the RANGE tab under AXES (Display 5). This tab is where we get to change the actual tick mark values. We do want a CUSTOM AXIS RANGE, so check the box. Our minimum value is 1997 (MIN) because that was the first year of data and our maximum value is 2012 (MAX) because that was the last year of data. The custom tick values should be in TICK SEQUENCE and we can now tell SAS that the sequence should start with 1997 and end with 2012 and the tick marks should be in increments of three. In this case, we choose three because our data are three years apart.

From the dropdown menu by AXIS under the AXES tab, select the Y-axis for editing (see arrow in Display 6). Because we already have the RANGE tab open under AXES, we will edit the values for the Y-axis first (Display 6). We want a CUSTOM AXIS RANGE. The MIN value should be 0 and the MAX value should be 60. We also want to specify the TICK SEQUENCE. The sequence of tick marks should start with 0 and end with 60 and should be in an increment of 5. If we select the GENERAL tab under AXES, we can give our Y-axis a label – “Percent (%) of Stroke Discharges by Age”. Select the LABEL tab under AXES and choose 18 for the FONT SIZE. Select the VALUE tab under AXES and choose 18 for the FONT SIZE. Choose OK.




Display 6. Plot Properties Change the Range of Y-Axis Values

We have now completed setting up the first scatter plot and the relevant information for the X and Y axes of the graph. We can now add our other plots to the graph. The order in which the additional plots are added does not matter. To add the regression plot that goes along with our first scatter plot, we will drag and drop a REGRESSION plot from the PLOT LAYERS section of the ELEMENTS window (Display 7). You may need to scroll down in the PLOT LAYERS section to find the correct plot. In the ASSIGN DATA window, you'll want to make sure that you are fitting the regression to an existing plot (checkbox). Choose the “scatter age group 1” plot from the available drop down menu for PLOT. You will remember that this is what we renamed the first scatter plot. We can rename our regression line in the ASSIGN DATA window if we wish. Choose OK.



Display 7. Insert Regression Plot from PLOT LAYERS


We need to update this regression plot's properties so we go to PLOT PROPERTIES by right clicking on the graph. Change the COLOR of the line to black, the PATTERN to DASH-DOT-DOT, and then change the LEGEND LABEL to "1-4 Years of Age" (Display 8).



Display 8. Regression Plot Properties


We need to create additional scatter plots and regression plots for the other three age groups but do not need to make any additional edits to the axes. To add an additional scatter plot, you will drag and drop a scatter plot from the PLOT LAYERS section of the ELEMENTS window onto your graph. Name your plots when you assign the data for and then go to PLOT PROPERTIES to change the properties of that each graph. Alternatively, you can add all the graphs then change the properties for each graph at once. You can find the same options as were used in the code under Method 1 such as the option to make markers STARFILLED in the menus.

Because we did not want to incorporate a title into the graph we are making, we can right click on the graph where it reads “Type in your title” and select REMOVE TITLE. We can also remove the footnote at the bottom of the graph. To insert the legend for the graph, go to INSERT on the menu bar and select GLOBAL LEGEND. In the box that comes up, select only the regression plots and make sure that the labels for the legend are how you want them to appear (Display 9). Choose OK.



Display 9. Select Entries for the Global Legend

Right-click on the legend and select LEGEND PROPERTIES. Change the FONT SIZE to 18, uncheck the OUTLINE box, select BOTTOM as the legend POSITION, select ACROSS as the legend LAYOUT, and enter 2 for the WRAP SIZE (Display 10). Choose OK. You will notice that the FONT SIZE option only allows for the changes in the LEGEND TITLE, not in the actual entries in the LEGEND. This is easily modifiable in the code we can get from ODS Graphics Designer.



Display 10. Global Legend Properties

After you finish creating all eight plots and the legend in ODS Graphics Designer, your result is Figure 2.




Figure 2. Resulting Graphic from ODS Graphics Designer

This graphic is not perfect yet but can be saved directly from ODS Graphics Designer as a TIFF or PNG file. If PNG is selected, the maximum available resolution is 600 DPI. Both this issue (for resolution) and the issue of changing the font size can be changed by altering the code that is available from the ODS Graphics Designer. To get to this code, go to VIEW on the menu bar and select CODE. A new window will open with code. Copy this code and paste it into a new program window in SAS 9.4 (or whatever version you are using). Save the graph you were working on in ODS Graphics Designer as a .sgd file and close ODS Graphics Designer. In the code you pasted in SAS 9.4, find the line that starts with DISCRETELEGEND under PROC TEMPLATE.

```
discretelegend '1-4 Years of Age' '5-9 Years of Age' '10-14 Years of Age'
'15-18 Years of Age' / opaque=true border=false halign=center valign=center
displayclipped=true across=1 order=rowmajor titleattrs=(size=18 );
```

Tell SAS that you want to change the font size of the values in the legend by adding VALUEATTRS = (SIZE = 18) before the semicolon in that code.

```
discretelegend '1-4 Years of Age' '5-9 Years of Age' '10-14 Years of Age'
'15-18 Years of Age' / opaque=true border=false halign=center valign=center
displayclipped=true across=1 order=rowmajor titleattrs=(size=18 )
VALUEATTRS = (SIZE = 18);
```

If you run your code, you will get a resulting graphic that looks like Figure 1. To save the graphic as 1200 DPI, use the same ODS statements that were used in Method 1.

```
ODS HTML DPI = 1200;
ODS PDF DPI = 1200 FILE = "C:\tiff\graph_new.pdf";
```

```

ODS RTF DPI = 1200 FILE = "C:\tiff\graph_new.doc";

proc sgrender data=WORK.TRENDDATA template=sgdesign;
dynamic _YEAR="YEAR" _AGEPER1A="AGEPER1" _YEAR2="YEAR" _AGEPER2A="AGEPER2"
_YEAR3="YEAR" _AGEPER3A="AGEPER3" _YEAR4="YEAR" _AGEPER4A="AGEPER4";
run;

ODS PDF CLOSE;
ODS RTF CLOSE;

```

If we run this code, we get the graphic that we created in ODS Graphics Designer as a 1200 DPI file that looks like Figure 1.

CONCLUSION

There are several tools and methods to create graphics for professional use in SAS. The PROC SGPLOT and ODS statements are excellent tools to create these graphics for any type of presentation from a slideshow to a PDF to a print journal. It is important to know what type of graphics each ODS option can create and display. It is also important to know how to change the details of each aspect of the graph to ensure that it can be read and viewed as intended. For users that prefer a menu-driven option with limited syntax, the ODS Graphics Designer is a very sufficient alternative to a code only method. Despite the sometimes tedious nature of creating graphics in SAS for professional uses, it is a welcome alternative to other software as it can create very detailed and sharp looking images.

REFERENCES

1. Slaughter, Susan J and Delwiche, Lora D. 2010. "Using PROC SGPLOT for Quick High-Quality Graphs." *Proceedings of the SAS Global Forum 2010*. Available at <http://support.sas.com/resources/papers/proceedings10/154-2010.pdf>.
2. SAS Institute Inc. "Sample 48138: Remove the border around the axis area and make the background white in graphics produced by PROC SGPLOT". October 26, 2012. Available at <http://support.sas.com/kb/48/138.html>.
3. SAS Institute Inc. "Setting the Resolution of Your Graphic". Available at <http://support.sas.com/documentation/cdl/en/graphref/65389/HTML/default/viewer.htm#n19kyqdftrtxqwc10x79xxtv18xh.htm>.
4. SAS Institute Inc. "Using Fonts with Universal Printers and SAS/GRAPH Devices". Available at <http://support.sas.com/documentation/cdl/en/lrcon/69852/HTML/default/viewer.htm#n04qo8tw2712qln1mzugjoorkj7j.htm>.
5. SAS Institute Inc. "PROC SGPLOT Concepts". Available at <http://support.sas.com/documentation/cdl/en/grstatproc/62603/HTML/default/viewer.htm#a003199399.htm>.
6. SAS Institute Inc. "Graph Size and Resolution". Available at http://support.sas.com/documentation/cdl/en/statug/68162/HTML/default/viewer.htm#statug_odsgraph_sect046.htm.
7. SAS Institute Inc. "Usage Note 31184: "Java virtual machine exception" message issued with ODS Graphics and large data sets." *SAS Knowledge Base*. August 31, 2010. Available at <http://support.sas.com/kb/31/184.html>.
8. Baker, C., Grant, A. M., George, M. G., Grosse, S. D., & Adamkiewicz, T. V. (2015). Contribution of Sick Cell Disease to the Pediatric Stroke Burden Among Hospital Discharges of African-Americans—United States, 1997–2012. *Pediatric Blood & Cancer*, 62(12), 2076–2081. doi:10.1002/pbc.25655
9. SAS Institute Inc. "SAS 9.4 ODS Graphics Designer: User's Guide, Third Edition". Available at <http://support.sas.com/documentation/cdl/en/grstatdesignug/68403/HTML/default/viewer.htm#titlepage.htm>.

Download code for the graphic: <https://github.com/dr baker11/SASGF17.git>

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Charlotte Baker
Florida Agricultural and Mechanical University Institute of Public Health
1515 S. Martin Luther King Jr Blvd
Tallahassee, FL 32307
charlotte.m.h.baker@gmail.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.